

Algoritmia

ADA 06: Algoritmos iterativos

Descripción: Utilizando la herramienta PSeInt realice los siguientes algoritmos mediante Pseudocódigo y diagramas de flujo.

Notas:

- Deberá exportar su código a un html y posteriormente pegarlo en word en el ejercicio correspondiente, esto para que el formato a colores se mantenga.
- Deberá exportar sus digramas de flujo al formato de imagen .png y posteriormente pegarlo en word en el ejercicio correspondiente, esto para que el formato a colores se mantenga.
- Recuerde lo siguiente:
 - o Utilizar el perfil UADY
 - o Escribir comentarios a su código.
 - o Utilizar nombres de variables de acuerdo a lo que vaya a almacenar.

1. Se requiere un algoritmo para obtener la edad promedio de un grupo de N alumnos mediante la utilización de un ciclo “Mientras”. Realice el diagrama de flujo y el pseudocódigo.

Algoritmo edadPromedio

//Definir Variables

Definir promedio **Como Real**;

Definir cantidadAlumnos, edadAlumno, numAlumno, edades **Como Entero**;

numAlumno = 1;

edades=0;

//Solicitamos información del usuario

Escribir "Ingresa la cantidad de Alumnos";

Leer cantidadAlumnos;

Mientras numAlumno ≤ cantidadAlumnos **Hacer**

//Pedimos la edad de numAlumno

Escribir "Ingresa la edad del alumno ", numAlumno, ":";

Leer edadAlumno;

edades = edades + edadAlumno;

numAlumno = numAlumno + 1;

Fin Mientras


promedio = edades/cantidadAlumnos;

Escribir "El promedio de edad de los ", cantidadAlumnos, " alumnos
es: ", promedio;

FinAlgoritmo



Algoritmia



Algoritmia

2. Se requiere un algoritmo para obtener la edad promedio de un grupo de N alumnos mediante la utilización de un ciclo Repite. Realice el diagrama de flujo y el pseudocódigo.

```
Algoritmo edadPromedio
  //Definir Variables
  Definir promedio Como Real;
  Definir cantidadAlumnos, edadAlumno, numAlumno, edades Como Entero;
  numAlumno = 1;
  edades=0;

  //Solicitamos información del usuario
  Escribir "Ingresa la cantidad de Alumnos";
  Leer cantidadAlumnos;

  Mientras numAlumno ≤ cantidadAlumnos Hacer
    //Pedimos la edad de numAlumno
    Escribir "Ingresa la cantidad del alumno ", numAlumno, ":";
    Leer edadAlumno;
    edades = edades + edadAlumno;


    numAlumno = numAlumno + 1;
  Fin Mientras

  promedio = edades/cantidadAlumnos;

  Escribir "El promedio de los ", cantidadAlumnos, " es: ", promedio;
FinAlgoritmo
```



Algoritmia



Algoritmia

3. Se requiere un algoritmo para obtener la edad promedio de un grupo de N alumnos mediante la utilización de un ciclo Desde. Representelo con el pseudocódigo y el diagrama de flujo.

Algoritmo edadPromedio

//Definimos Variables

Definir edad, sumaEdades, nAlumnos, nAlumno **Como Entero**;

Definir promedio **Como Real**;

sumaEdades = 0;

//Pedimos el numero de alumnos

Escribir "Ingrese la cantidad de alumnos";

Leer nAlumnos;

//empezamos desde el alumno numero 1 hasta llegar con nAlumnos

Para nAlumno = 1 **Hasta** nAlumnos **Con Paso 1 Hacer**

//Pedimos la edad del alumno nAlumno

Escribir "Ingresa la edad del alumno ", nAlumno, ":";

Leer edad;

sumaEdades = sumaEdades + edad;

Fin Para

//Calculamos el promedio

promedio = sumaEdades/nAlumnos;


//Escribimos el promedio

Escribir "La edad promedio de los ", nAlumnos, " alumnos es: ",
promedio;

FinAlgoritmo



Algoritmia



Algoritmia

4. Se requiere un algoritmo para obtener la estatura promedio de un grupo de personas, cuyo número de miembros se desconoce, el ciclo debe efectuarse siempre y cuando se tenga una estatura registrada. Representelo mediante el pseudocódigo y el diagrama de flujo.

Algoritmo estaturaPromedio

```
//Definimos variables
Definir estatura, estaturas, promedio Como Real;
Definir cantidad Como Entero;
cantidad = 0;
estaturas = 0;
//Solcitamos estaturas hasta que ya no ingresen valores
Repetir
    Escribir "Ingresa tu estatura en metros:";
    Leer estatura;
    estaturas = estaturas + estatura;

    //sumamos uno a la cantidad para saber la cantidad de estaturas
    ingresadas
    cantidad = cantidad + 1;
Hasta Que estatura = 0

    //Le restamos 1 a la cantidad para no contar la vez que no se ingreso
    un valor
    cantidad = cantidad - 1;


    //Calculams el promedio
    promedio = estaturas/cantidad;

    //Escribimos el promedio de las estaturas
    Escribir "El promedio de las ", cantidad, " estaturas ingresadas es:
    ", promedio, "cm.";
```

FinAlgoritmo



Algoritmia



Algoritmia


5. Realice un algoritmo para obtener la tabla de multiplicar de un entero K comenzando desde el 1. Represéntelo mediante diagrama de flujo y pseudocódigo.

```
Algoritmo tablaK
  //Definimos las variables
  Definir k, n, resultado Como Entero;
  //igualamos a 1 a la variable b para iniciat con 1 la tablas
  n = 1;
  //Pedimos el valor k
  Escribir "Ingresa el número para saber su tabla de multiplicar: ";
  Leer k;

  //Obtener la tabla de multiplicar de K desde 1 hasta 12
  Mientras n ≤ 12 Hacer
    resultado = k * n;
    Escribir k, " * ", n, " = ", resultado ;
    n = n + 1;
  Fin Mientras
FinAlgoritmo
```



Algoritmia



Algoritmia

6. Se requiere un algoritmo para determinar cuánto ahorrará una persona en un año, si al final de cada mes deposita variables cantidades de dinero; además, se requiere saber cuánto lleva ahorrado cada mes. Representelo mediante el pseudocódigo y el diagrama de flujo.

Algoritmo ahorro

```
//Definimos variables
Definir contador Como Real;
Definir ahorroMes, ahorroTotal Como Real;
ahorroTotal = 0 ;
Definir mes Como Caracter;
Dimension mes[13];

mes[1] = "Enero";
mes[2] = "Febrero";
mes[3] = "Marzo";
mes[4] = "Abril";
mes[5] = "Mayo";
mes[6] = "Junio";
mes[7] = "Julio";
mes[8] = "Agosto";
mes[9] = "Septiembre";
mes[10] = "Octubre";
mes[11] = "Noviembre";
mes[12] = "Diciembre";


//Repetimos 12 veces para completar el año
Para contador = 1 Hasta 12 Con Paso 1 Hacer
    //Solicitamos el ahorro del mes
    Escribir "Ingresa el ahorro del mes de ", mes[contador], ":";
    Leer ahorroMes;
    ahorroTotal = ahorroTotal + ahorroMes;
    Si contador < 12 Entonces
        Escribir "El ahorro que llevas es: $", ahorroTotal;
    Fin Si
Fin Para

//Escribimos el ahorro anual
Escribir "El ahorro del año es: $", ahorroTotal;

FinAlgoritmo
```



Algoritmia



Algoritmia

7. Realice un algoritmo para generar e imprimir los números primos que se encuentran entre 0 y 100. Recordar que un número primo es aquel que sólo es divisible entre 1 y él mismo. Representelo mediante el pseudocódigo y el diagrama de flujo.

Algoritmo numerosPrimos

```
//Definimos variables
Definir contador, contadorPrimos Como Entero;
Definir esPrimo Como Logico;

Escribir "Numeros primos: ";

//Iniciamos en 2 porque el 1 nunca será primo
Para contadorPrimos = 2 Hasta 100 Con Paso 1 Hacer
    esPrimo = Verdadero;

    //no comprobamos el 1 porque un primo si es divisible por 1
    //Tampoco consideramos cuando los numeros son iguales pues
    //un numero primo es divisible por sí mismo
    Para contador = 2 Hasta contadorPrimos-1 Con Paso 1 Hacer


        //Si el resto es 0 entonces es divisible por algún número
        //entonces no es un numero primo
        Si contadorPrimos%contador == 0 Entonces
            esPrimo = Falso;
        Fin Si
    Fin Para

    //Imprimimos el número si es primo
    Si esPrimo Entonces
        Escribir contadorPrimos;
    Fin Si
Fin Para

FinAlgoritmo
```



Algoritmia



Algoritmia

8. Realice un algoritmo para generar N elementos de la sucesión de Fibonacci (0, 1, 1, 2, 3, 5, 8, 13,...). Represéntelo mediante el pseudocódigo y el diagrama de flujo.

```
Algoritmo fibonacci
  //Definimos variables
  Definir n, contador Como Entero;

  //Si los definimos como entero llegamos hasta n = 47
  //Definir umAnt1, numAnt2, num Como Entero;

  //Se definen como real para obtener una mayor capacidad de numeros
  //Aunque en n = 80 empieza a hacer calculos incorrectos
  //es decir solo se imprimen correctamente hasta n = 79
  Definir numAnt1, numAnt2, num Como Real;

  //La sucesion fibonacci inicia con el 0 y el 1
  numAnt1 = 0;
  numAnt2 = 1;


  //Pedimos la cantidad de numeros a Imprimir
  Escribir "Ingresa la cantidad de miembros de la sucesion fibonacci a
  imprimir:";
  Leer n;

  Si n>0 Entonces
    Escribir "1: ", numAnt1;
    Si n>1 Entonces
      Escribir "2: ", numAnt2;
      Para contador = 3 Hasta n Con Paso 1 Hacer
        num = numAnt1 + numAnt2;
        numAnt1 = numAnt2;
        numAnt2 = num;
      Escribir contador, ": ", num;
    Fin Para
  Fin Si
Fin Si

FinAlgoritmo
```



Algoritmia



Algoritmia

9. Una empresa tiene el registro de las horas que trabaja diariamente un empleado durante la semana (seis días) y requiere determinar el total de éstas, así como el sueldo que recibirá por las horas trabajadas. Represente en pseudocódigo y diagrama de flujo.

Algoritmo horasSueldo

//Definimos variables

Definir numDia, horas, totalHoras **Como Entero**;

Definir sueldoHora, sueldoTotal **Como Real**;

numDia = 1;

totalHoras = 0;

//Pedimos el pago por hora

Escribir "Ingresa el pago por hora: ";

Leer sueldoHora;

//Pedimos las horas de trabajo por día

Mientras numDia \leq 6 **Hacer**

Escribir "Ingresa las horas trabajadas el día ", numDia, ": ";

Leer horas;

 totalHoras = totalHoras + horas;

 numDia = numDia + 1;

Fin Mientras

//Calculamos el sueldoTotal

sueldoTotal = sueldoHora * totalHoras;


//Imprimimos los resultados

Escribir "Las horas trabajadas son: ", totalHoras;

Escribir "El sueldo total es: ", sueldoTotal;

FinAlgoritmo

Algoritmia



Algoritmia

10. Un empleado de la tienda “Tiki Taka” realiza N ventas durante el día, se requiere saber cuántas de ellas fueron mayores a \$1000, cuántas fueron mayores a \$500 pero menores o iguales a \$1000, y cuántas fueron menores o iguales a \$500. Además, se requiere saber el monto de lo vendido en cada categoría y de forma global. Realice un algoritmo que permita determinar lo anterior y represéntelo mediante el diagrama de flujo y el pseudocódigo.

Algoritmo tikiTaka

```
//Definimos variables
Definir contador, nVentas, mayorMil, mayorQuinien, demas Como Entero;
Definir ventaTemp, totalVentas, totalMayorMil, totalMayorQuinien,
totalDemas Como Real;
mayorMil = 0;
mayorQuinien = 0;
demas = 0;
totalMayorMil = 0;
totalMayorQuinien = 0;
totalDemas = 0;

//Solicitamos el numero de ventas del día
Escribir "¿Cuántas ventas se hicieron en el día?";
Leer nVentas;

//Solicitamos las "N ventas" y la clasificamos
Para contador = 1 Hasta nVentas Con Paso 1 Hacer
    Escribir "Ingresa el monto de la venta ", contador, ":";
    Leer ventaTemp;

    //Clasificamos
    Si ventaTemp > 1000 Entonces
        totalMayorMil = totalMayorMil + ventaTemp;
        mayorMil = mayorMil + 1;
    SiNo
        Si ventaTemp > 500 Entonces
            totalMayorQuinien = totalMayorQuinien + ventaTemp;
            mayorQuinien = mayorQuinien + 1;
        SiNo
            totalDemas = totalDemas + ventaTemp;
            demas = demas + 1;
    Fin Si
Fin Si
Fin Para


//Imprimimos resultados
Escribir "# Ventas TikiTaka:";
Escribir "-----";
Escribir "Ventas >$1000: ", mayorMil;
Escribir "Total de ventas: $", totalMayorMil;
Escribir "-----";
Escribir "Ventas >$500, <=$1000: ", mayorQuinien;
Escribir "Total de ventas: $", totalMayorQuinien;
```



Algoritmia

```
Escribir "-----";  
Escribir "Ventas <$500: ", demas;  
Escribir "Total de ventas: $", totalDemas;  
Escribir "-----";
```

FinAlgoritmo



Algoritmia

11. Una persona adquirió un producto para pagar en 20 meses. El primer mes pagó \$10, el segundo \$20, el tercero \$40 y así sucesivamente. Realice un algoritmo para determinar cuánto debe pagar mensualmente y el total de lo que pagó después de los 20 meses y represéntelo mediante el diagrama de flujo y el pseudocódigo.

Algoritmo pago20meses

//Definimos variables

Definir pagoMes, pagoTotal, contador **Como Entero**;

contador = 1;

//iniciamos con 10 debido a que así lo indica el problema

pagoMes = 10;

pagoTotal = pagoMes;

Escribir "Pago mes 1: ", pagoMes;

Repetir

 contador = contador + 1;

 pagoMes = pagoMes * 2;

//se agrega lo que se paga cada mes al pago total

 pagoTotal = pagoTotal + pagoMes;

Escribir "Pago mes ", contador, ": ", pagoMes;


Hasta Que contador = 20

//Escribimos el pago total

Escribir "Pago total: ", pagoTotal;

FinAlgoritmo

Algoritmia



Algoritmia

12. Una empresa les paga a sus empleados con base en las horas trabajadas en la semana. Para esto, se registran los días que laboró y las horas de cada día. Realice un algoritmo para determinar el sueldo semanal de N trabajadores y además calcule cuánto pagó la empresa por los N empleados.

Algoritmo sueldoTrabajadores

```
//Definimos variables
Definir idia, dias, cantTrabajadores, nTrabajador Como Entero;
Definir sueldoHora, sueldoTotalTrabajador, sueldoTotalTrabajadores,
horas, sumaHorasTrabajador Como Real;
sueldoTotalTrabajadores = 0;

//solicitamos sueldo y tabajadores al usuario
Escribir "Ingresa la cantidad de trabajadores: ";
Leer cantTrabajadores;
Escribir "Ingresa el sueldo por hora: ";
Leer sueldoHora;

Para nTrabajador = 1 hasta cantTrabajadores Hacer
    Escribir "Ingresa la cantidad de días que ha trabajado el
empleado ", nTrabajador, ":";
    Leer dias;
    sumaHorasTrabajador = 0;
    Para idia = 1 hasta dias Hacer
        Escribir "Ingresa la cantidad de horas que trabajo el
trabajador ", nTrabajador, " el dia ", idia, ":";
        Leer horas;
        sumaHorasTrabajador = horas + sumaHorasTrabajador;
    FinPara
    sueldoTotalTrabajador = sueldoHora * sumaHorasTrabajador;
    //sumar el sueldo de cada uno de los trabajadores
    sueldoTotalTrabajadores = sueldoTotalTrabajadores +
sueldoTotalTrabajador;


    Escribir "El trabajador ", nTrabajador, " recibe un sueldo de $",
sueldoTotalTrabajador;
FinPara

//Escribimos el pago total
Escribir "El pago total de los ", cantTrabajadores, " trabajadores es
de: $", sueldoTotalTrabajadores;

FinAlgoritmo
```



Algoritmia



Algoritmia

13. La cadena de tiendas de autoservicio “El mandilón” cuenta con sucursales en C ciudades diferentes de la República, en cada ciudad cuenta con T tiendas y cada tienda cuenta con N empleados, asimismo, cada una registra lo que vende de manera individual cada empleado, cuánto fue lo que vendió cada tienda, cuánto se vendió en cada ciudad y cuánto recaudó la cadena en un solo día. Realice un algoritmo para determinar lo anterior y represéntelo mediante un diagrama de flujo, utilizando el ciclo apropiado.

Algoritmo elMandilon

```
//Definimos variables
Definir numCiudad, cCiudades, numTienda, tTiendas, numEmpleado,
nEmpleados, numVenta, xVentas Como Entero;
Definir venta, totalVentaCiudad, totalVentaTienda, ventaTotal Como
Real;

//Pedimos numero de ciudades
Escribir "Ingresa el numero de ciudades: ";
Leer cCiudades;

//Calculamos la venta global
ventaTotal = 0;
Para numCiudad = 1 Hasta cCiudades Con Paso 1 Hacer

    //pedimos numero de tienda
    Escribir "Ingresa el numero de tiendas en la ciudad #",
numCiudad, ": ";
    Leer tTiendas;

    //Calculamos el total de venta de la cCiudad
    totalVentaCiudad = 0;
    Para numTienda = 1 Hasta tTiendas Con Paso 1 Hacer

        //Pedimos el numero de empleados
        Escribir "Ingresa el numero de empleados de la tienda #",
numTienda, ": ";
        Leer nEmpleados;

        //Calculamos el totalVentaTienda
        totalVentaTienda = 0;
        Para numEmpleado = 1 Hasta nEmpleados Con Paso 1 Hacer

            //Pedimos la venta de numEmpleado
            Escribir "Ingresa la venta del empleado numero #",
numEmpleado, ": ";
            Leer venta;
            totalVentaTienda = totalVentaTienda + venta;

Fin Para
```

Algoritmia

```
//Escribimos el total de la venta de la tienda
Escribir "El total de venta de la tienda es: $",
totalVentaTienda;
totalVentaCiudad = totalVentaCiudad + totalVentaTienda;

Fin Para

//Escribir el totalVentaCiudad
Escribir "La venta total de la ciudad #", numCiudad, "es: $",
totalVentaCiudad;

ventaTotal = ventaTotal + totalVentaCiudad;


Fin Para

//Escribimos el total recaudado en el día
Escribir "El dinero recaudado total es de: $", ventaTotal;

FinAlgoritmo
```



Algoritmia



Algoritmia

14. Un profesor tiene un salario inicial de \$1500, y recibe un incremento de 10 % anual durante 6 años. ¿Cuál es su salario al cabo de 6 años? ¿Qué salario ha recibido en cada uno de los 6 años? Realice el algoritmo y represente la solución mediante el diagrama de flujo y pseudocódigo.

Algoritmo salarioProfesor


```
//Definimos variables
Definir contador Como Entero;
Definir salario, salarioTotal Como Real;
contador = 1;

//El salario inicia con 1500 como lo indica el problema
salario = 1500;
Escribir "El salario inicial es $", salario;
Escribir "Al año gano: $", salario*12;
salarioTotal = salario * 12;
Repetir
    //Calculamos el anadido
    salario = salario * 1.10;
    Escribir "El salario del profesor despues de ", contador, " año/s
es $", salario;
    Escribir "Al año gano: $", salario * 12;
    contador = contador + 1;
    salarioTotal = salarioTotal + salario * 12;
Hasta Que contador > 6;

//Al final del año el salario
Escribir "Al final del 6to año habrá ganado un total de: $",
salarioTotal;
FinAlgoritmo
```



Algoritmia



Algoritmia

15. Realice el algoritmo para determinar cuánto pagará una persona que adquiere N artículos, los cuales están de promoción. Considere que si su precio es mayor o igual a \$200 se le aplica un descuento de 15%, y si su precio es mayor a \$100 pero menor a \$200, el descuento es de 12%; de lo contrario, sólo se le aplica 10%. Se debe saber cuál es el costo y el descuento que tendrá cada uno de los artículos y finalmente cuánto se pagará por todos los artículos obtenidos. Represente la solución mediante el diagrama de flujo y el pseudocódigo.

Algoritmo nArticulos

```
//Definimos variables
Definir n, nContador Como Entero;
Definir pago, pagoDescuento, precio, descuento, precioDescuento Como
Real;
pago = 0;
pagoDescuento = 0;
//Pedimos la cantidad de articulos
Escribir "Ingresa la cantidad de artículos a comprar";
Leer n;

Para nContador = 1 Hasta n Con Paso 1 Hacer
    //Le pedimos el precio del producto nContador
    Escribir "Ingresa el precio del articulo ", nContador;
    Leer precio;

    // Se determina que descuento se le aplicara
    //Y se guarda el precio con descuento en otra variable
    Si precio ≥ 200 Entonces
        descuento = precio * .15;
    SiNo
        Si precio > 100 Entonces
            descuento = precio * .12;
        SiNo
            descuento = precio * .10;
        Fin Si
    Fin Si

    //Calcular precio con descuento
    precioDescuento = precio - descuento;

    Escribir "El precio del producto con descuento es: $",
precioDescuento;


    //Calculamos pagos totales
    pago = pago + precio;
    pagoDescuento = pagoDescuento + precioDescuento;
Fin Para

//Escribimos los pagos totales
Escribir "El pago total es: $", pago;
Escribir "El pago total con descuento es: $", pagoDescuento;

FinAlgoritmo
```



Algoritmia



Algoritmia

16. Realice un algoritmo y represéntelo mediante un diagrama de flujo para obtener una función exponencial, la cual está dada por

$$e^x = 1 + \frac{x}{1!} + \frac{x^2}{2!} + \frac{x^3}{3!} + \dots,$$

```
Algoritmo funcionExponencial
//Definimos variables
Definir resultado, elevacion, factorial, x Como Real;
Definir i, j Como Entero;
resultado = 1;

//Pedimos el valor de x
Escribir "Este algoritmo calcula el valor de e^x, Ingresa el valor
para x:";
Leer x;


//Hacemos el calculo iterativo hasta 100 pues el resultado es
bastante acertado con este valor
Para i = 1 Hasta 100 Con Paso 1 Hacer
    //elevacion
    elevacion = x;
    Para j = 2 Hasta i Con Paso 1 Hacer
        elevacion = elevacion * x;
    Fin Para
    //factorial
    j = 1;
    factorial = 1;
    Mientras j ≤ i Hacer
        factorial = factorial * j;
        j = j + 1;
    Fin Mientras

    //calculamos resultado
    resultado = resultado + elevacion/factorial;
Fin Para

//Escribimos el resultado
Escribir "e^", x, " = ", resultado;
FinAlgoritmo
```




Algoritmia



Algoritmia

17. Un vendedor ha realizado N ventas y desea saber cuántas fueron por 10,000 o menos, cuántas fueron por más de 10,000 pero por menos de 20,000, y cuánto fue el monto de las ventas de cada una y el monto global. Realice un algoritmo para determinar los totales. Represente la solución mediante diagrama de flujo, pseudocódigo.

Algoritmo nVentas

```
//Definimos variables
Definir n, nContador, menorIgual10k, mayor10kmenor20k Como Entero;
Definir venta, montoTotal, montomenorIgual10k, montomayor10kmenor20k
Como Real;
menorIgual10k = 0;
mayor10kmenor20k = 0;
montomenorIgual10k = 0;
montomayor10kmenor20k = 0;
montoTotal = 0;

//Pedimos la cantidad de ventas
Escribir "Ingresa la cantidad de ventas";
Leer n;

Para nContador = 1 Hasta n Con Paso 1 Hacer
    //Le pedimos el precio del producto nContador
    Escribir "Ingresa el total de la venta ", nContador;
    Leer venta;

    // Se determina a que grupo pertenece
    Si venta ≤ 10000 Entonces
        menorIgual10k = menorIgual10k + 1;
        montomenorIgual10k = montomenorIgual10k + venta;
    SiNo
        Si venta < 20000 Entonces
            mayor10kmenor20k = mayor10kmenor20k + 1;
            montomayor10kmenor20k = montomayor10kmenor20k + venta;
        Fin Si
    Fin Si


    //Calcular el montoTotal de las n ventas
    montoTotal = montoTotal + venta;
Fin Para

//Escribimos los resultados
Escribir "Hubo un total de ", menorIgual10k, " ventas que fueron por
10,000 o menos con un monto de: $", montomenorIgual10k;
Escribir "Hubo un total de ", mayor10kmenor20k, " ventas que fueron
por mas de 10,000 y menos de 20,000 con un monto de: $",
montomayor10kmenor20k;
Escribir "El monto total de las ",n , " ventas fue de $", montoTotal;

FinAlgoritmo
```



Algoritmia



Algoritmia

18. Realice un algoritmo para obtener el seno de un ángulo y represéntelo mediante diagrama de flujo y pseudocódigo.

$$\text{Sen } x = (x - x^3/3! + x^5/5! - x^7/7! + \dots)$$

Algoritmo seno

```
//Definimos variables
Definir angulo, fSeno, factorial, elevacion Como Real;
Definir i, j Como Entero;
Definir resta Como Logico;
Definir tipoAngulo Como Caracter;
resta = Falso;
fSeno = 0;
factorial = 1;

//Leer el angulo;
Escribir "Ingresa el angulo: ";
Leer angulo;

//Identificamos el ingreso del angulo
Escribir "¿El angulo esta en grados o radianes? (Ingresa R para radianes, G para grados.)";
Leer tipoAngulo;

//Hacemos la conversion del angulo en caso que sea necesario
Si tipoAngulo == 'G' Entonces
    angulo = angulo / 57.2958;
Fin Si
//calcular el seno
Para i = 1 Hasta 41 Con Paso 2 Hacer

    //calcular potencia
    Para j = 1 Hasta i Con Paso 1 Hacer
        Si j = 1 Entonces
            elevacion = angulo;
        SiNo
            elevacion = elevacion * angulo;
        Fin Si
    Fin Para

    //calcular factorial
    factorial = 1;
    Para j = 2 Hasta i Con Paso 1 Hacer
        factorial = factorial * j;
    Fin Para

    //Determinar si toca resta o suma
    Si resta Entonces
        elevacion = elevacion * (-1);
        resta = Falso;
```



Algoritmia

```
SiNo
    resta = Verdadero;
Fin Si

    //calculo iterativo del seno de un ángulo
    fSeno = fSeno + elevacion/factorial;


Fin Para

    //Escribimos el valor del seno del angulo
    Escribir "Seno = " fSeno;

FinAlgoritmo
```



Algoritmia



Algoritmia

19. El banco “Bandido de peluche” desea calcular para cada uno de sus N clientes su saldo actual, su pago mínimo y su pago para no generar intereses. Además, quiere calcular el monto de lo que ganó por concepto interés con los clientes morosos. Los datos que se conocen de cada cliente son: saldo anterior, monto de las compras que realizó y pago que depositó en el corte anterior. Para calcular el pago mínimo se considera 15% del saldo actual, y el pago para no generar intereses corresponde a 85% del saldo actual, considerando que el saldo actual debe incluir 12% de los intereses causados por no realizar el pago mínimo y \$200 de multa por el mismo motivo. Realice el algoritmo correspondiente y representelo mediante diagrama de flujo y pseudocódigo.

Algoritmo SaldoIntereses

```
//Definimos Variables
//Actuales
Definir saldoActual, compra, pagoMinimo, pagoSinIntereses Como Real;
//Relacion Previa
Definir saldoAnterior, pagoAnterior, pagoMinimoAnt,
pagoSinInteresesAnt Como Real;
// nClientes y contador
Definir nClientes, contador Como Entero;

//Pedimos el total de clientes
Escribir "Ingrese el total de clientes";
Leer nClientes;

Para contador = 1 Hasta nClientes Con Paso 1 Hacer

    Escribir "----- Cliente ", contador, " -----";
    //solicitamos saldo y deposito
    Escribir "Ingrese el saldo del cliente ", contador, ":";
    Leer saldoAnterior;
    Escribir "Ingrese la cantidad que depósito anteriormente:";
    Leer pagoAnterior;
    Escribir "Ingrese el total de su compra:";
    Leer compra;
    Escribir "Saldo anterior: $", saldoAnterior;

    //Calculamos pagos minimos y saldo actual
    pagoMinimoAnt = saldoAnterior*.15;
    pagoSinInteresesAnt = saldoAnterior*.85;
    saldoActual = saldoAnterior + compra - pagoAnterior;;

    //Agregamos intereses o multa
    Si pagoAnterior < pagoMinimoAnt Entonces
        saldoActual = saldoActual * 1.12 + 200;
    SiNo
        Si pagoAnterior < pagoSinInteresesAnt Entonces
            saldoActual = saldoActual * 1.12;
        FinSi
    FinSi
```


Algoritmia

```
//calculamos pagos minimos nuevos
pagoMinimo = saldoActual*.15;
pagoSinIntereses = saldoActual*.85;

//Devolvemos al usuario su informacion
Escribir "Saldo actual es $", saldoActual;
Escribir "Para no generar intereses debe pagar: $",
pagoSinIntereses, ".";
Escribir "Para no generar multa por atraso debe pagar: $",
pagoMinimo, ".";

Fin Para
```

FinAlgoritmo



Algoritmia

20. En 1961, una persona vendió las tierras de su abuelo al gobierno por la cantidad de \$1,500. Suponga que esta persona ha colocado el dinero en una cuenta de ahorros que paga 15% anual. ¿Cuánto vale ahora su inversión? $P(1+i)^n$. Realice un algoritmo para obtener este valor y represéntelo mediante diagrama de flujo y pseudocódigo.

```
Algoritmo inversion
//Definimos variables
Definir total Como Real;

//total = 1500;
//Definir contador Como Entero;
//Iniciamos con 1962, pues ahí ya generará su primer 15% de ganancia
// Para contador = 1962 Hasta 2020 Con Paso 1 Hacer
//     total = total* 1.15;
// Fin Para
//Escribimos el total de su inversión

//Usando la formula  $P(1+i)^n$  tenemos un código más sencillo

total = ((1.15)^(2020-1961)) * 1500;
//Escribimos el total
Escribir total;
FinAlgoritmo
```

